
Visual Basic Editor (VBE) is the tool used to create, modify, and maintain Visual Basic For Applications (VBA) procedures and modules in MS Office applications.   VBA gives you the ability to modify workbooks and worksheets through VBA, as if you were going through the Excel interface.  VBA also allows you to modify VBA components and code modules, as if you were going through the VBE interface.    This page applies only to Excel97 and above.  It does not apply to Excel95 or previous versions.  

This pages describes a few of the objects, methods, and properties of the VBE that you can manipulate from VBA.  In Excel97, these objects, methods, and properties are not described in the normal VBA help files.  You need to open the file called VEENOB3.hlp.  This file many not have been installed on your system when you installed the VBA help files and Office97.  You can find it in the MoreHelp folder on your Excel or Office CD.  You many want to have a macro, assigned to a menu item or a shortcut key to easily display this file. 

Sub ShowVBEHelp()

Shell "c:\windows\winhelp.exe veenob3.hlp", vbNormalFocus

End Sub 

In Excel 2000 and 2002, these topics are included in the standard VBA help files.  

Before using these procedures, you'll need to set a reference in VBA to the VBA Extensibility library.  In the VBA editor, go to the Tools menu, choose the References item, and put a check next to "Microsoft Visual Basic For Applications Extensibility" library.   This enables VBA to find the definitions of these objects.  If you are using Excel97, this library will appear in the References list without a version number:  "Microsoft Visual Basic For Applications Extensibility".  If you are using Excel 2000 or 2002, it will appear with a version number: "Microsoft Visual Basic For Applications Extensibility 5.3".  It is very important that you reference the proper library.  If you reference the wrong library, you will receive "Type Mismatch" errors.  If you don't reference the extensibility library at all, you will receive "User Defined Type Not Defined Error".

For  information about programming the menus in the VBE, see the Adding Menus To The VBA Editor page.

Note: An additional level of security was added in Excel 2002.  To manipulate the VBA Project objects as described here, you'll have to change your security setting.  Go to the Tools menu, choose Macros, then Security. Click the "Trusted Sources" tab, and put a check next to the "Trust access to Visual Basic Project". 

NOTE: In all versions of Excel, the VBProject must not be protected.  If it is, these procedures will fail.  In Excel 2002, you must have "Trust Access To Visual Basic Project" enabled.  To enable this setting, go to the Tools menu in Excel, choose Macros, Security, then the "Trusted Sources" tab, and put a check next to "Trust Access To Visual Basic Project".  Otherwise, you will get errors.

Also, you may get unpredictable results if you attempt to modify a code module's code from that same module.  That is, having code in Module1 modify the contents of Module1.  I recommend that you do not do this.

VBE Objects 

We'll be using three of these objects in our code:  

VBProject                This is the entire set of VBA modules and references associated with a workbook.  

VBComponent       This is the individual component within a VBProject.  For example, a UserForm and a standard code module are each a VBComponent.  The VBComponents collection contains each existing VBComponent object.  

CodeModule            This object represents the actual code contained in a VBComponent.  For example, when you enter code into Module1, you're entering code into the CodeModule object of the VBComponent named "Module1".   

We'll be programmatically "navigating" to these components through the Workbook object.  You can also get to these components by going through the Application.VBE object path, but we won't be doing this.   

There are various types of VBComponents, identified by the Type property of the VBComponent object.  

         Type Constant Description 

vbext_ct_ClassModule 

This is a class module, used to create your own objects.  We won't be using these here. 

vbext_ct_Document 

This is the component for a worksheet, chart sheet, or ThisWorkbook. 

vbext_ct_MSForm 

This is the component for a UserForm. 

vbext_ct_StdModule 

This is the component for a standard code module.  Most of our procedures will work with these components. 

Getting A Reference To An Object 

The first step in programming to the VBE is to get a reference to object you need to work with.   

VBProject                

Dim VBProj As VBProject

Set VBProj = ThisWorkbook.VBProject 

VBComponent                

Dim VBComp As VBComponent

Set VBComp = ThisWorkbook.VBProject.VBComponents("Module1") 

CodeModule                

Dim VBCodeMod As CodeModule

Set VBCodeMod = ThisWorkbook.VBProject.VBComponents("Module1").CodeModule 

In all of the examples in this page, we'll be working with the ThisWorkbook object -- working with the VBA components in the workbook which contains the code.  Of course, you can work with any open workbook, by using ActiveWorkbook or Workbooks("SomeBook.xls"). 

Adding A Module To A Workbook  

The procedure below will add a new module named "NewModule" to ThisWorkbook. 

Sub AddModule()

Dim VBComp As VBComponent

Set VBComp = ThisWorkbook.VBProject.VBComponents.Add(vbext_ct_StdModule)

VBComp.Name = "NewModule"

Application.Visible = True

End Sub 

When you run this code from Excel while the VBE is open, you will be taken to the new module's code module, and the macro will terminate.  When you run this code while the VBE is not open, your Excel application will be visible, but will not have focus.  The  statement returns focus back to the Excel application. 

Deleting A Module From A Workbook 

The procedure below will delete the module named "NewModule" from ThisWorkbook.  

Sub DeleteModule()

Dim VBComp As VBComponent

Set VBComp = ThisWorkbook.VBProject.VBComponents("NewModule")

ThisWorkbook.VBProject.VBComponents.Remove VBComp

End Sub 

You cannot delete the ThisWorkbook code module, or a sheet code module, or a chart code module. 

Adding A Procedure To A Module 

The procedure below will add a new procedure called "MyNewProcedure" to the module named "NewModule" in ThisWorkbook. 

Sub AddProcedure()

Dim VBCodeMod As CodeModule

Dim LineNum As Long

Set VBCodeMod = ThisWorkbook.VBProject.VBComponents("NewModule").CodeModule

With VBCodeMod

    LineNum = .CountOfLines + 1

    .InsertLines LineNum, _

"Sub MyNewProcedure()" & Chr(13) & _

" Msgbox ""Here is the new procedure"" " & Chr(13) & _

"End Sub"

End With

Application.Run "MyNewProcedure"

End Sub 

Pay attention to the way in which the .InsertLines method is called.  The entire procedure is passed as one argument -- a string with embedded Chr(13) characters for the line breaks.  The code statement 

Application.Run "MyNewProcedure" 

will run the new procedure.  You must use Application.Run rather than calling the procedure directly in order to prevent compile-time errors.  This method will work only if you are adding code to another code module.  If you are adding code a the same code module, you must use an Application.OnTime method, so that control is returned to Excel, and the module can be recompiled and reloaded.  Using Application.OnTime may have some synchronizations problems, so you should avoid calling a procedure that you've just added to the same code module without allowing all VBA procedures to come to an end.   

Application.OnTime Now,"NewProcedureName" 

Creating An Event Procedure 

The CodeModule object has a method called CreateEventProc that you can use to create an event procedure in class module, a sheet module, or the ThisWorkbook module. The advantage of CreateEventProc over InsertLines is that CreateEventProc  will automatically insert the complete procedure declaration, including all of the correct parameters. CreateEventProc  returns the line number on which the procedure begins, so once you've called  CreateEventProc , add one to the result and use this with InsertLines  to insert the body of the event procedure.  For example, the code below creates a Workbook_Open procedure containing a Msgbox statement in the ThisWorkbook module of the Active Workbook.

Dim StartLine As Long

With ActiveWorkbook.VBProject.VBComponents("ThisWorkbook").CodeModule

    StartLine = .CreateEventProc("Open", "Workbook") + 1

    .InsertLines StartLine, _

    "Msgbox ""Hello World"",vbOkOnly"

End With

Deleting A Procedure From A Module 

The procedure below will delete the procedure called "MyNewProcedure" from the module named "NewModule" in ThisWorkbook. 

Sub DeleteProcedure()

Dim VBCodeMod As CodeModule

Dim StartLine As Long

Dim HowManyLines As Long

Set VBCodeMod = ThisWorkbook.VBProject.VBComponents("NewModule").CodeModule

With VBCodeMod

    StartLine = .ProcStartLine("MyNewProcedure", vbext_pk_Proc)

    HowManyLines = .ProcCountLines("MyNewProcedure", vbext_pk_Proc)

    .DeleteLines StartLine, HowManyLines

End With

End Sub 

Deleting All Code From A Module 

The procedure below will delete all code from a module name "NewModule".

Sub DeleteAllCodeInModule()

Dim VBCodeMod As CodeModule

Dim StartLine As Long

Dim HowManyLines As Long

Set VBCodeMod = ThisWorkbook.VBProject.VBComponents("NewModule").CodeModule

With VBCodeMod

    StartLine = 1

    HowManyLines = .CountOfLines

   .DeleteLines StartLine, HowManyLines

End With

End Sub 

Listing All Modules In A Workbook 

The procedure below will list, in a message box, all of the modules in ThisWorkbook.  It uses a function called CompTypeToName to get a string describing the type of module.   The function CompTypeToName is listed below.  

Sub ListModules()

Dim VBComp As VBComponent

Dim Msg As String

For Each VBComp In ThisWorkbook.VBProject.VBComponents

    Msg = Msg & VBComp.Name & " Type: " & CompTypeToName(VBComp) & Chr(13)

Next VBComp

MsgBox Msg

End Sub 

Function CompTypeToName(VBComp As VBComponent) As String

Select Case VBComp.Type

    Case vbext_ct_ActiveXDesigner

        CompTypeToName = "ActiveX Designer"

    Case vbext_ct_ClassModule

        CompTypeToName = "Class Module"

    Case vbext_ct_Document

        CompTypeToName = "Document"

    Case vbext_ct_MSForm

        CompTypeToName = "MS Form"

    Case vbext_ct_StdModule

        CompTypeToName = "Standard Module"

    Case Else

End Select

End Function 

Listing All Procedures In A Module 

The procedure below will list, in a message box, all of the procedures in a standard code module called "SaveModule" in ThisWorkbook.  Procedures are listed in the order in which they appear in the CodeModule object.    

Sub ListProcedures()

Dim VBCodeMod As CodeModule

Dim StartLine As Long

Dim Msg As String

Dim ProcName As String

Set VBCodeMod = ThisWorkbook.VBProject.VBComponents("SaveModule").CodeModule

With VBCodeMod

    StartLine = .CountOfDeclarationLines + 1

    Do Until StartLine >= .CountOfLines

        Msg = Msg & .ProcOfLine(StartLine, vbext_pk_Proc) & Chr(13)

        StartLine = StartLine + _

          .ProcCountLines(.ProcOfLine(StartLine, _    

           vbext_pk_Proc), vbext_pk_Proc)

    Loop

End With

MsgBox Msg

End Sub 

Also see Code Modules And Code Names for more information about the CodeName property of VBComponents.

Exporting All Modules In A Project 

The procedure below will list export all of the modules in a workbook to text files. It will save the files in the same folder as the workbook.  This can be useful for saving a backup copy of your VBA, or for transferring VBA code from one project to another. 

Sub ExportAllVBA()

Dim VBComp As VBIDE.VBComponent

Dim Sfx As String

For Each VBComp In ActiveWorkbook.VBProject.VBComponents

    Select Case VBComp.Type

        Case vbext_ct_ClassModule, vbext_ct_Document

           Sfx = ".cls"

        Case vbext_ct_MSForm

           Sfx = ".frm"

        Case vbext_ct_StdModule

           Sfx = ".bas"

        Case Else

           Sfx = ""

    End Select

    If Sfx <> "" Then

       VBComp.Export _

          Filename:=ActiveWorkbook.Path & "\" & VBComp.Name & Sfx

    End If

Next VBComp

End Sub

Deleting All VBA Code In A Project 

The procedure below will delete all the VBA code in a project.  You should use this procedure with care, as it will permanently delete the code.  Standard modules, user forms, and class modules will be removed, and code within the ThisWorkbook module and the sheet modules will be deleted.  You may want to export the VBA code, using the procedure above, before deleting the VBA code. 

Sub DeleteAllVBA()

Dim VBComp As VBIDE.VBComponent

Dim VBComps As VBIDE.VBComponents

Set VBComps = ActiveWorkbook.VBProject.VBComponents

For Each VBComp In VBComps

   Select Case VBComp.Type

      Case vbext_ct_StdModule, vbext_ct_MSForm, _

            vbext_ct_ClassModule

         VBComps.Remove VBComp

      Case Else

         With VBComp.CodeModule

            .DeleteLines 1, .CountOfLines

         End With

   End Select

Next VBComp

End Sub

Copying Modules Between Projects 

There isn't a single method to copy modules from one VBProject to another.  Instead, you have to export the module from one project, and then import it into another.   The following procedure will copy Module1 from Book2 to Book1.   

Sub CopyOneModule()

Dim FName As String

With Workbooks("Book2")

    FName = .Path & "\code.txt"

    .VBProject.VBComponents("Module1").Export FName

End With

Workbooks("book1").VBProject.VBComponents.Import FName

End Sub 

Just change "Module1" to the name of the module you want to copy.  If you want to copy all modules (except the ThisWorkbook and Sheet modules), you can use the following code.  

Sub CopyAllModules()

Dim FName As String

Dim VBComp As VBIDE.VBComponent

With Workbooks("Book2")

    FName = .Path & "\code.txt"

    If Dir(FName) <> "" Then

        Kill FName

    End If

    For Each VBComp In .VBProject.VBComponents

        If VBComp.Type <> vbext_ct_Document Then

           VBComp.Export FName

           Workbooks("book1").VBProject.VBComponents.Import FName

           Kill FName

        End If

    Next VBComp

End With

End Sub

Testing Existence Of A Module Or Procedure 

You can use the VBA Extensibility tools to determine whether a module exists, or a procedure exists in a module.  

Function ModuleExists(ModuleName As String) As Boolean

On Error Resume Next

ModuleExists = Len( _

ThisWorkbook.VBProject.VBComponents(ModuleName).Name) <> 0

End Function

Function ProcedureExists(ProcedureName As String, _ 

    ModuleName As String) As Boolean

On Error Resume Next

If ModuleExists(ModuleName) = True Then

    ProcedureExists = ThisWorkbook.VBProject.VBComponents(ModuleName) _

       .CodeModule.ProcStartLine(ProcedureName, vbext_pk_Proc) <> 0

End If

End Function

